Universidad del País Vasco
Euskal Herriko Unibertsitatea

Ejercicios resueltos

OCW 2018: *Utilizando Mathematica como apoyo al cálculo algebraico en los grados de Ingeniería*

Tema 4. Espacios vectoriales de dimensión finita

Equipo docente del curso
Martín Yagüe, Luis
Unzueta Inchaurbe, Aitziber
Arrospide Zabala, Eneko
García Ramiro, María Begoña
Soto Merino, Juan Carlos
Alonso González, Erik

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I

EJERCICIOS DEL TEMA 4. ESPACIOS VECTORIALES DE DIMENSIÓN FINITA

Ejercicio nº1

Enunciado

Determine si los siguientes subconjuntos del espacio vectorial $\mathbb{R}^4$ son libres o ligados. En caso de ser ligados especifique una combinación lineal.

- a) $S = \{(1, 0, 2, 1), (0, 2, -1, -1), (2, 6, 1, -1)\}$
- b) $T = \{(1, 2, 1, 0), (1, -1, 0, 1), (2, 1, 1, -1)\}$

Resolución

- a) $S = \{(1, 0, 2, 1), (0, 2, -1, -1), (2, 6, 1, -1)\}$
  - Introducción del subconjunto S como una matriz:

```
S = {u1 = {1, 0, 2, 1}, u2 = {0, 2, -1, -1}, u3 = {2, 6, 1, -1}}; S // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 2 & 1 \\ 0 & 2 & -1 & -1 \\ 2 & 6 & 1 & -1 \end{pmatrix}$$

  - Se utiliza la función **MatrixRank[*m*]** para calcular el rango de la matriz: $r(S) = 2 < n^o\ de\, vectores$

```
MatrixRank[S]
```

2

  - Como $r(S) = 2 < n^o\ de\, vectores$, el sistema es ligado
  - A continuación, se calcula la combinación lineal; para ello, se estudia si los vectores $\vec{u}_1$ y $\vec{u}_2$ son libres:

```
MatrixRank[{u1, u2}]
```

2

  - $\vec{u}_1$ y $\vec{u}_2$ son libres puesto que el rango de la matriz que forman es dos
  - Se determina una combinación lineal:

```
Solve[a * u1 + b * u2 == u3]
```

{{a → 2, b → 3}}

  - En este caso, se tiene: $\boxed{\vec{u}_3 = 2\,\vec{u}_1 + 3\,\vec{u}_2}$

- b) $T = \{(1, 2, 1, 0), (1, -1, 0, 1), (2, 1, 1, -1)\}$
  - Introducción del subconjunto T como una matriz

```
T = {{1, 2, 1, 0}, {1, -1, 0, 1}, {2, 1, 1, -1}}; T // MatrixForm
```

$$\begin{pmatrix} 1 & 2 & 1 & 0 \\ 1 & -1 & 0 & 1 \\ 2 & 1 & 1 & -1 \end{pmatrix}$$

- Se utiliza la función **MatrixRank[*m*]** para calcular el rango de la matriz

```
MatrixRank[T]
```

```
3
```

- Como el rango de la matriz es 3, igual al número de vectores, el sistema es libre

Ejercicio nº2

Enunciado

Sea el subespacio vectorial $S = \mathcal{L}(\{2x^3 - x + 1, 2x^3 + 2x, -2x^3 + 4x - 2, -3x^3 + 6x - 3\})$ del espacio vectorial $\mathbb{P}_3(x)$.

- a) Calcule una base de S.
- b) Obtenga unas ecuaciones paramétricas e implícitas del subespacio S.
- c) Determine si los polinomios $p(x) = x^2 + 2x$ y $q(x) = 3x - 1$ pertenecen al subespacio vectorial S.
- d) Calcule, si es posible, las coordenadas de los polinomios $p(x) = x^2 + 2x$ y $q(x) = 3x - 1$ utilizando la base obtenida en el primer apartado.

Resolución

- a) Calcule una base de S.
  - Introducción de los polinomios:

```
p1[x_] = 2 x^3 - x + 1; p2[x_] = 2 x^3 + 2 x; p3[x_] = -2 x^3 + 4 x - 2; p4[x_] = -3 x^3 + 6 x - 3;
```

  - Generación de una matriz con las coordenadas de los polinomios dispuestas por filas: función **CoefficientList[*poly*, *var*]**

```
cm = CoefficientList[{p1[x], p2[x], p3[x], p4[x]}, x]
```

```
{{1, -1, 0, 2}, {0, 2, 0, 2}, {-2, 4, 0, -2}, {-3, 6, 0, -3}}
```

  - Matriz equivalente escalonada por filas: función **RowReduce[*m*]**

```
cmr = RowReduce[cm]; cmr // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 3 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

  - Las filas no nulas de la matriz son los coeficientes de los polinomios de una base de S:

$$\boxed{B_S = \{3x^3 + 1,\ x^3 + x\}}$$

- b) Obtenga las ecuaciones paramétricas e implícitas del subespacio S.
  - Obtención de unas ecuaciones paramétricas:

```
α * cmr[[1]] + β * cmr[[2]]
```

```
{α, β, 0, 3 α + β}
```

  - Solución: $\boxed{S = \{p(x) = dx^3 + cx^2 + bx + a \ / \ a = \alpha,\ b = \beta,\ c = 0,\ d = 3\alpha + \beta \quad \forall\, \alpha, \beta \in \mathbb{R}\}}$
  - Obtención de las ecuaciones implícitas:

```
ie = {cmr[[1]], cmr[[2]], {a, b, c, d}}; ie // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 3 \\ 0 & 1 & 0 & 1 \\ a & b & c & d \end{pmatrix}$$

```
Det[ie[[{1, 2, 3}, {1, 2, 3}]]] == 0
```

c == 0

```
Det[ie[[{1, 2, 3}, {1, 2, 4}]]] == 0
```

-3 a - b + d == 0

- Solución: $\boxed{S = \{p(x) = dx^3 + cx^2 + bx + a \ / \ c = 0,\ d - 3a - b = 0 \ \ \forall\ a, b, d \in \mathbb{R}\}}$

- c) Determine si los polinomios $p(x) = x^2 + 2x$ y $q(x) = 3x - 1$ pertenecen al subespacio vectorial S.
  - Se introduce el polinomio $p(x) = x^2 + 2x$

```
p[x_] = x^2 + 2 x
```

$2x + x^2$

  - Se genera una matriz con las coordenadas de los polinomios que forman la base de S junto con las coordenadas del polinomio $p(x)$ y se estudia su rango

```
M = {cmr[[1]], cmr[[2]], CoefficientList[p[x], x, 4]}; MatrixRank[M]
```

3

  - Como el rango es tres, igual al número de filas, los tres vectores (fila) son libres: $\boxed{p(x) \notin S}$
  - Análogamente, se realizan los cálculos para $q(x) = 3x - 1$

```
q[x_] = 3 x - 1; MatrixRank[{cmr[[1]], cmr[[2]], CoefficientList[q[x], x, 4]}]
```

2

  - Como el rango es dos, menor que el número de filas, los tres vectores (fila) no son libres: $\boxed{q(x) \in S}$

- d) Calcule, si es posible, las coordenadas de los polinomios $p(x) = x^2 + 2x$ y $q(x) = 3x - 1$ utilizando la base obtenida en el primer apartado.
  - No es posible calcular las coordenadas del polinomio $p(x)$ dado que $p(x) \notin S$
  - Se calculan las coordenadas del polinomio $q(x)$:

```
Solve[CoefficientList[q[x], x, 4] == a * cmr[[1]] + b * cmr[[2]]]
```

{{a → -1, b → 3}}

  - Entonces: $\boxed{q(x) = (-1, 3)_{B_S}}$

Ejercicio nº3

Enunciado

Sea el subespacio vectorial $S = \mathcal{L}\left(\left\{\begin{pmatrix} 1 & 0 \\ -1 & -1 \end{pmatrix}, \begin{pmatrix} 1 & 1 \\ 1 & 1 \end{pmatrix}, \begin{pmatrix} 4 & 3 \\ 2 & 2 \end{pmatrix}, \begin{pmatrix} -4 & -5 \\ -6 & -6 \end{pmatrix}\right\}\right) \subset \mathbb{M}_{2\times 2}(\mathbb{R})$.

- a) Calcule una base de S.
- b) Obtenga unas ecuaciones paramétricas e implícitas del subespacio S.
- c) Sea el subespacio vectorial $T = \left\{ \begin{pmatrix} a & b \\ c & d \end{pmatrix} \Big/ c = d \wedge b = 0 \right\}$. Calcule $S \cap T$.

Resolución

- a) Calcule una base de S.
  - Introducción de las matrices dadas:

```
u1 = {{1, 0}, {-1, -1}}; u2 = {{1, 1}, {1, 1}}; u3 = {{4, 3}, {2, 2}}; u4 = {{-4, -5}, {-6, -6}};
```

  - Generación de una matriz con las coordenadas de las mismas por filas: función **Flatten[*list*]**

```
cm = {Flatten[u1], Flatten[u2], Flatten[u3], Flatten[u4]};
```

  - Obtención de una matriz equivalente escalonada por filas

```
cmr = RowReduce[cm]; cmr // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & -1 & -1 \\ 0 & 1 & 2 & 2 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

  - Por tanto, una base del subespacio vectorial S es: $\boxed{B_S = \left\{ \begin{pmatrix} 1 & 0 \\ -1 & -1 \end{pmatrix}, \begin{pmatrix} 0 & 1 \\ 2 & 2 \end{pmatrix} \right\}}$

- b) Obtenga unas ecuaciones paramétricas e implícitas del subespacio S.
  - Obtención de unas ecuaciones paramétricas:

```
α * cmr[[1]] + β * cmr[[2]]
```

$\{\alpha, \beta, -\alpha + 2\beta, -\alpha + 2\beta\}$

  - Por lo que: $\boxed{S = \left\{ \begin{pmatrix} a & b \\ c & d \end{pmatrix} \Big/ a = \alpha,\ b = \beta,\ c = -\alpha + 2\beta,\ d = -\alpha + 2\beta \quad \forall\, \alpha, \beta \in \mathbb{R} \right\}}$

  - Obtención de unas ecuaciones implícitas:

```
ie = {cmr[[1]], cmr[[2]], {a, b, c, d}}; ie // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & -1 & -1 \\ 0 & 1 & 2 & 2 \\ a & b & c & d \end{pmatrix}$$

```
Det[ie[[{1, 2, 3}, {1, 2, 3}]]] == 0
```

$a - 2b + c == 0$

```
Det[ie[[{1, 2, 3}, {1, 2, 4}]]] == 0
```

$a - 2b + d == 0$

  - Entonces: $\boxed{S = \left\{ \begin{pmatrix} a & b \\ c & d \end{pmatrix} \Big/ a - 2b + c = 0 \wedge a - 2b + d = 0 \right\}}$

  - También: $\boxed{S = \left\{ \begin{pmatrix} a & b \\ c & d \end{pmatrix} \Big/ c = 2b - a \wedge c = d \right\}}$

- c) Sea el subespacio vectorial $T = \left\{ \begin{pmatrix} a & b \\ c & d \end{pmatrix} \Big/ c = d \wedge b = 0 \right\}$. Calcule $S \cap T$.
    - Se resuelve el sistema formado por las ecuaciones implícitas de ambos subespacios:

```
Solve[{a - 2 b + c == 0, a - 2 b + d == 0, c == d, b == 0}]
```

```
{{b → 0, c → -a, d → -a}}
```

- Por lo que el subespacio es: $\boxed{S \cap T = T = \left\{ \begin{pmatrix} a & 0 \\ -a & -a \end{pmatrix} \Big/ a \in \mathbb{R} \right\} \subset S}$

Ejercicio nº4

Enunciado

Sean $B_1 = \{(1, 0), (1, 2)\}$ y $B_2 = \{(0, 1), (-2, -1)\}$ dos bases del espacio vectorial $\mathbb{R}^2$.

- a) Sea el vector $\vec{x} = (5, 6)$, calcule sus coordenadas en la base B_1.
- b) Sea el vector $\vec{x} = (5, 6)$ calcule sus coordenadas en la base B_2.
- c) Estudie la relación existente entre las coordenadas obtenidas en los apartados anteriores.

Resolución

- a) Sea el vector $\vec{x} = (5, 6)$, calcule sus coordenadas en la base B_1.
    - Definición del vector $\vec{x} = (5, 6)$ y los vectores que forman la base B_1

```
x = {5, 6}; b11 = {1, 0}; b12 = {1, 2};
```

- Se expresa el vector como combinación lineal de los vectores de la base y se resuelve el sistema resultante:

```
Solve[x == a * b11 + b * b12]
```

```
{{a → 2, b → 3}}
```

- Coordenadas del vector en la base B_1: $\boxed{\vec{x} = (2, 3)_{B_1}}$

- b) Sea el vector $\vec{x} = (5, 6)$ calcule sus coordenadas en la base B_2.
    - Definición de los vectores que forman la base B_2

```
b21 = {0, 1}; b22 = {-2, -1};
```

- Se expresa el vector como combinación lineal de los vectores de la base y se resuelve el sistema resultante:

```
Solve[x == a * b21 + b * b22]
```

$\left\{\left\{a \to \frac{7}{2}, b \to -\frac{5}{2}\right\}\right\}$

- Coordenadas del vector en la base B_2: $\boxed{\vec{x} = \left(\frac{7}{2}, -\frac{5}{2}\right)_{B_2}}$

- c) Estudie la relación existente entre las coordenadas obtenidas en los apartados anteriores.
    - La matriz de cambio de base relaciona las coordenadas de un vector en dos bases diferentes

- Se calcula la matriz de cambio de base de la base B_1 a la B_2
- Se obtienen las coordenadas de los vectores de la base B_1 en la base B_2

```
s1 = Solve[b11 == a * b21 + b * b22]
```

$\left\{\left\{a \to -\frac{1}{2}, b \to -\frac{1}{2}\right\}\right\}$

```
s2 = Solve[b12 == a * b21 + b * b22]
```

$\left\{\left\{a \to \frac{3}{2}, b \to -\frac{1}{2}\right\}\right\}$

- Se tiene: $\boxed{(1,0)=\left(-\frac{1}{2}, -\frac{1}{2}\right)_{B_2}}$ $\boxed{(1,2)=\left(\frac{3}{2}, -\frac{1}{2}\right)_{B_2}}$
- Colocando dichas coordenadas por columnas se obtiene la matriz de cambio de base:

```
cb = Transpose[{{a, b} /. s1[[1]], {a, b} /. s2[[1]]}]; cb // MatrixForm
```

$\begin{pmatrix} -\frac{1}{2} & \frac{3}{2} \\ -\frac{1}{2} & -\frac{1}{2} \end{pmatrix}$

- Solución: $\boxed{M_{B_1 \longrightarrow B_2} = \begin{pmatrix} -\frac{1}{2} & \frac{3}{2} \\ -\frac{1}{2} & -\frac{1}{2} \end{pmatrix}}$
- Comprobación: se multiplica la matriz calculada por las coordenadas del vector $\vec{x}$ en la base B_1 y se obtienen sus coordenadas en la base B_2

```
cb.{2, 3}
```

$\left\{\frac{7}{2}, -\frac{5}{2}\right\}$

Ejercicio nº5

Enunciado

Sean los subespacios vectoriales $S = \{p(x) \in \mathbb{P}_3(x) \,/\, p(1) = p'(1) = 0\}$ y $T = \mathcal{L}(\{1+x, x^2\})$.

- a) Calcule una base de S.
- b) Calcule una base de T.
- c) Calcule $S+T$.
- d) Obtenga una base de $\mathbb{P}_3(x)$ completando la base de $S+T$.

Resolución

- a) Calcule una base de S.
  - Obtención de una base de S calculando unas ecuaciones implícitas a partir de un vector genérico $p(x) \in S$:

```
p[x_] = d * x^3 + c * x^2 + b * x + a;
s = Solve[{p[1] == 0, p'[1] == 0}]
```

{{c → -3 a - 2 b, d → 2 a + b}}

- Expresión general de un polinomio $p(x) \in S$:

```
g[x_] = p[x] /. s[[1]]
```

```
a + b x + (-3 a - 2 b) x^2 + (2 a + b) x^3
```

- Obtención de un sistema generador:

```
p1[x_] = g[x] /. {a → 1, b → 0}
```

```
1 - 3 x^2 + 2 x^3
```

```
p2[x_] = g[x] /. {a → 0, b → 1}
```

```
x - 2 x^2 + x^3
```

- Comprobación de que el sistema es libre colocando sus coordenadas por filas en una matriz:

```
MatrixRank[CoefficientList[{p1[x], p2[x]}, x, 4]]
```

```
2
```

- Sistema libre y generador, por tanto, es una base de S:
  $\boxed{B_S = \{\, p_1(x) = 2\,x^3 - 3\,x^2 + 1,\ p_2(x) = x^3 - 2\,x^2 + x \,\}}$

- b) Calcule una base de T.
  - Definición de los polinomios del sistema generador de T :

```
q1[x_] = 1 + x; q2[x_] = x^2;
```

- Comprobación de que los polinomios son libres:

```
MatrixRank[CoefficientList[{q1[x], q2[x]}, x, 4]]
```

```
2
```

- Como el rango es dos, los dos polinomios son libres y forman una base:
  $\boxed{B_T = \{\, q_1(x) = x + 1,\ q_2(x) = x^2 \,\}}$

- c) Calcule $S + T$.
  - Se obtiene un sistema generador del subespacio $S + T$ uniendo las bases obtenidas en el apartado anterior:

$$S + T = \mathcal{L}\left(\left\{1 - 3\,x^2 + 2\,x^3,\ x - 2\,x^2 + x^3,\ 1 + x,\ \ x^2\right\}\right)$$

  - Para obtener una base de $S + T$ se comprueba que el sistema generador es libre:

```
MatrixRank[CoefficientList[{p1[x], p2[x], q1[x], q2[x]}, x, 4]]
```

```
4
```

- El sistema es libre y generador, por tanto, es una base de $S + T$:
  $\boxed{B_{S+T} = \left\{1 - 3\,x^2 + 2\,x^3,\ x - 2\,x^2 + x^3,\ 1 + x,\ \ x^2\right\}}$

- d) Obtenga una base de $\mathbb{P}_3(x)$ completando la base de $S + T$.
  - Dado que subespacio $dim(S + T) = 4 = dim\,(\mathbb{P}_3(x))$ el subespacio $S + T$ coincide con el espacio total: $S + T = \mathbb{P}_3(x)$
  - Entonces, la base pedida es: $\boxed{B_{\mathbb{P}_3(x)} = \left\{1 - 3\,x^2 + 2\,x^3,\ x - 2\,x^2 + x^3,\ 1 + x,\ \ x^2\right\}}$